

グラフ描画ソフト“gnuplot”の使い方

授業中に使用している高性能なグラフ描画ソフト“gnuplot”(フリーウェア)の使い方を紹介します。

ダウンロード&インストールの仕方：ネットでgnuplotで検索すると沢山のサイトがヒットします。多分最初に出てくる総本山のサイト <http://www.gnuplot.info/> 「gnuplot homepage」からフリーでダウンロードできます。使用法例も載っています。OSがWindowsなら最新版 version4.4.2 は、gp442win32.zip をダウンロードします (mac, unix 用もあります)。これを適当なフォルダーで解凍すると新たにgnuplotフォルダーが生成されます。その下部のbinaryフォルダーにある実行ファイル wgnuplot.exe をクリックすればgnuplotが起動出来ます。但し、gnuplot画面の初期フォントが非常に見にくいのでgnuplot画面上で右クリックして Choose Font で適当なフォントを選んで下さい。

グラフの表示の仕方の基本：gnuplot画面で最終行に >gnuplot と表示されているはずですが、その後にカーソルが点滅しているのです、そこから例えば

```
plot sin(x)
```

と入力してEnterすると別のウインドーが開いて関数 $y = \sin(x)$ のグラフが表示されます。但し、最初にグラフ表示する時に少し時間が掛ります。以下の注意を参照して下さい。

注意：version4.4.0以降のgnuplotをインストールして最初にグラフを表示する時、別のウインドーが開いて20秒~30秒グラフが表示されない場合があります。少しの時間我慢して下さい。実際、私がversion4.4.2をダウンロード&インストールした時にこのような状態でしたが30秒位待ったらやっとグラフが表示されました。グラフ表示のデフォルトのターミナルがversion4.4.0以前の'windows'から'wxt'に変更された影響と思われる。素人なので詳しくは知りません。2度目の表示からはあっと言う間に表示されます。あまりに長い間表示されない様なら、ダウンロード&インストールが失敗しているのかも知れません。何度か試してダメならversion4.4.0より前のversion4.0.0~4.2.6で試して下さい。

スクリプトファイル：gnuplotでは、コマンドを並べて書いたテキストファイルのスクリプトファイルとかバッチファイルと呼んでいます。gnuplotの画面で“file”をクリックし“open”をクリックして、作ったスクリプトファイルを選択するとgnuplotは書かれた順番にコマンドを逐次実行してグラフを表示します。gnuplot画面のカーソルにload'ファイル名'と入力してもOKです。スクリプトファイルはテキストファイルですから適当なテキストエディタで作れば良いのです。スクリプトファイルの拡張子は何でも良いのですが、.pltを使う事が多い様です。

例1. 周期 2π の関数 $f(x) = -\frac{x}{2}$ ($-\pi \leq x \leq \pi$) のフーリエ級数 $f(x) = \sum_{n=1}^{\infty} \frac{(-1)^n}{n} \sin(nx)$ を表示するスクリプトファイル rei1.plt を下に書きます。

```
set samples 512
set grid
set xrange [-4*pi:4*pi]
set xtics -4*pi,pi,4*pi
set yrange [-pi:pi]
set ytics -pi,pi/2,pi
f(n,x)=(n==0)?0:f(n-1,x)+(-1)**n*sin(n*x)/n
plot f(50,x)
```

この rei1.plt は $y = \sum_{n=1}^{50} \frac{(-1)^n}{n} \sin(nx)$ のグラフを表示するスクリプトファイルの一例です。少し解説すると下から2行目は

$$f(n, x) = \begin{cases} 0 & (n = 0 \text{ の時}) \\ f(n-1, x) + \frac{(-1)^n}{n} \sin(nx) & (\text{それ以外の時}) \end{cases}$$

で関数列 $\{f(n, x)\}_{n=0}^{\infty}$ を定義しています。

例2. が裏面にあります

例 2. フーリエ解析第 6 回演習の例題 :

関数 $y = y(x, t)$ についての波動方程式

$$\frac{\partial^2 y}{\partial t^2} = 9 \frac{\partial^2 y}{\partial x^2} \dots \textcircled{1} \quad (0 \leq x \leq 4, t \geq 0)$$

を次の境界条件 , 初期条件の下で解け .

$$\text{境界条件} \quad y(0, t) = y(4, t) = 0 \dots \textcircled{2} \quad (t \geq 0)$$

$$\text{初期条件} \quad y(x, 0) = \begin{cases} x & (0 \leq x \leq 2) \\ 4 - x & (2 < x \leq 4) \end{cases} \dots \textcircled{3}, \quad \frac{\partial y}{\partial t}(x, 0) = 0 \dots \textcircled{4}$$

の解

$$y = \frac{16}{\pi^2} \sum_{n=1}^{\infty} \frac{(-1)^{n-1}}{(2n-1)^2} \cos \left\{ \frac{3(2n-1)\pi t}{4} \right\} \sin \frac{(2n-1)\pi x}{4}$$

で時間のパラメータ t を $t = 0, \frac{1}{12}, \frac{2}{12}, \dots, \frac{32}{12} = \frac{8}{3}$ として $\frac{1}{12}$ 刻みで表示するスクリプトファイル 10en6-rei.plt を下に書きます .

```
set samples 512
set grid
set xrange [0:4]
set yrange [-3:3]
unset key
f(n,i,x) = (n==0) ? 0 : f(n-1,i,x) + (16/pi**2)*((-1)**(n-1))/(2*n-1)**2*cos(3*(2*n-1)*pi/4*i/12)*sin((2*n-1)*pi*x/4)
i=0
plot f(100,i,x)
load 'time.rot'
```

上の 10en6-rei.plt の最終行のコマンド load 'time.rot' は下のスクリプトファイル time.rot を読み込む物です :

```
i=i+1
pause 0.1
replot
if (i<32) reread
```

詳しい事は次の参考文献で勉強して下さい .

参考文献 :

- ・ gnuplot の精義 山本昌志 著 カットシステム
- ・ 使いこなす gnuplot (改定第 2 版) 矢吹道朗 監修 大竹敢 著 テクノプレス